I am curious whether Jews view Scripture (i.e. the Tanach) as "the word of God," and if so, in what way? Is it "inspired" by God? Is it a "product of the Spirit of God"? Is God the "Author"? [etc.]

In other words, what was God's role in the production of Scripture from a Jewish perspective?


Question: Preventing Superscript from being interpreted as Power when using Ctrl+^ shortcut?
Answer: Superscript is not interpreted as Power:



Presumably you are referring to what happens when you enter a power in superscript notation using the key combination Ctrl+6.  Mathematica is capable of representing both this power notation and a formatted plain Superscript.  In my opinion it is a failing that the power notation appears in the Typesetting menu while the latter is missing; if anything it should be the other way around I think.  Since there is no key combination for raw Superscript I propose using a palette or input alias:

Palette

You may enter a formatting template using a palette button which may be created with:

CreatePalette @ PasteButton @ Superscript[\[SelectionPlaceholder], \[Placeholder]]




Click that palette button to insert a template for plain superscript in the current Notebook.  Use Tab to move between fields.

Input alias

Either open the Option Inspector, select Global Preferences, type "InputAliases" to find the appropriate entry, and add this to the list of rules:

"sps" -&gt; TemplateBox[{"\[SelectionPlaceholder]", "\[Placeholder]"}, "Superscript"]


Or add it programmatically (run this only once):

AppendTo[
 CurrentValue[$FrontEnd, "InputAliases"],
 "sps" -&gt; TemplateBox[{"\[SelectionPlaceholder]", "\[Placeholder]"}, "Superscript"]
]


Now type: EscspsEsc to enter a template for plain Superscript.

(In version 10.0.0 the first field will not be automatically selected due to a bug; see Input Aliases in Mathematica 10.)



Bounty

A bounty was added for:


  [A] solution to map the Ctrl+^ keybinding to produce superscript instead of power.


To accomplish this first copy MenuSetup.tr from the installation directory to the matching path in your user directory and open the user copy for editing:

os = $OperatingSystem /. "Unix" -&gt; "X";

CopyFile @@ (
  FileNameJoin[{#, "SystemFiles", "FrontEnd", "TextResources", os, "MenuSetup.tr"}] &amp; /@
    {$InstallationDirectory, $UserBaseDirectory})

% // SystemOpen


Then within the user copy edit the appropriate MenuItem to read:

MenuItem["&amp;Superscript", 
 FrontEndExecute[{
   FrontEnd`SelectionMove[FrontEnd`InputNotebook[], All, Word], 
   FrontEnd`NotebookApply[FrontEnd`InputNotebook[], 
    TemplateBox[{"\[SelectionPlaceholder]", "\[Placeholder]"}, "Superscript"], 
    Placeholder]
 }],
 MenuKey["6", Modifiers -&gt; {"Control"}]
]


Restart Mathematica.

You may now enter raw Superscript by using Ctrl+6 where 6 is the number-line six above the alphabetic keyboard.

You can still enter Power notation by using Ctrl+Shift+6 or Ctrl+Keypad-6, the latter assuming that Num Lock is on.



For those you prefer the reverse behavior you can instead copy and edit KeyEventTranslations.tr and change the Item:

Item[KeyEvent["^", Modifiers -&gt; {Control}], "Superscript"]


to:

Item[KeyEvent["^", Modifiers -&gt; {Control}], 
  FrontEndExecute[{
     FrontEnd`SelectionMove[FrontEnd`InputNotebook[], All, Word], 
     FrontEnd`NotebookApply[FrontEnd`InputNotebook[], 
      TemplateBox[{"\[SelectionPlaceholder]", "\[Placeholder]"}, "Superscript"], 
      Placeholder]
  }]
]


Now Ctrl+6 is Power and Ctrl+Shift+6 is raw Superscript.  However the Typesetting menu item remains misleadingly named Superscript so I would personally change that to Power if adopting this binding.

Also see:


Can I modify Ctrl-- to insert Indexed expressions?